# Maintainer: Thomas Dziedzic < gostrc at gmail >
# Contributor: Joeny Ang <ang(dot)joeny(at)gmail(dot)com>

pkgname=simple-scan
pkgver=3.8.0
pkgrel=1
pkgdesc='Simple scanning utility'
arch=('i686' 'x86_64')
url='http://launchpad.net/simple-scan'
license=('GPL3')
depends=('sane' 'dconf' 'gtk3' 'colord' 'hicolor-icon-theme')
makedepends=('yelp-tools' 'intltool')
install=simple-scan.install
source=(https://launchpad.net/$pkgname/${pkgver:0:3}/$pkgver/+download/$pkgname-$pkgver.tar.xz)
sha256sums=('8df71e5f9e5db5c53217cb56b0e23b58dfe65bc174e873ad3ee576fbeccadc22')

build() {
  cd $pkgname-$pkgver
  ./configure \
    --prefix=/usr \
    --disable-schemas-compile
  make
}

package() {
  cd $pkgname-$pkgver
  make DESTDIR="$pkgdir" install
}
